
Disclaimer: 
 
Trotz sorgfältiger Tests kann für die Korrektheit der Programme und Funktionen keine Gewähr übernommen werden. 
Besonders im Bereich mit Monitorauflösungen wird immer wieder von der möglichen  Zerstörung von Monitoren bei 
falschen Einstellungen gewarnt, besonders bei Röhrenbildschirmen (CRT).   Ich tue dies hiermit auch. 
Aus genau diesem Grund wurde auch die „arbeitsintensive“ Batchvariante gewählt, bei  der jeder Monitor zuerst mit 
seinen Werten eingetragen werden muss. Dadurch wird eine Gefahr sehr minimiert. 
„Sachdienliche“ Rückmeldungen jeglicher Art sind erwünscht! 
 
 
Für diese Arbeit wurden folgende Freeware-Tools verwendet: 
 

- Devcon.exe  von Microsoft:  Ein Befehlszeilendienstprogramm als Alternative zum Geräte-Manager 
o DevCon :  http://support.microsoft.com/kb/311272/de 

- Dccmd.exe  von Stefan Tucker:  Setzt Bildschirmauflösungen auch für mehrere Monitore 
o DisplayChanger:   http://www.12noon.com/displaychanger.htm 

- WizApp.exe  von Dion Nicolaas:   Gibt Skripts eine Benutzeroberfläche 
o Wizard's Apprentice:  http://wizapp.sf.net  

 
Herzlichen Dank an die oben genannten Programmierer und sonstigen Helfer.  
 

Helmut Pajnik,  für das Team der IT-Regionalbetreuer des Landes Vorarlberg  
 

http://support.microsoft.com/kb/311272/de
http://www.12noon.com/displaychanger.htm
http://wizapp.sf.net/

1. Hintergrundbild und Auflösung passend zum Monitor 

1.1. Problemstellung 

Da mittlerweile  in den Schulen unterschiedliche Generationen von Monitoren mit  maximalen Auflösungen zwischen 
1024x768 bis 1920x1080 verwendet werden, stimmt nach dem Ausrollen des Standardimages bei vielen PCs weder die 
Auflösung noch die Größe unseres  Hintergrundbildes. Dies muss dann manuell pro PC korrigiert werden, sofern die 
Benutzer nicht vor einem matschigen Bild sitzen sollen. 

1.2. Zielvorgabe 

Beim Start eines PCs soll automatisch entsprechend dem angeschlossenen Monitor  

 auf eine vorgegebene (optimale) Auflösung umgestellt und  

 das dazu passend große Hintergrundbild verwendet werden. 

 Beim Anmelden eines Schülers sollen die Icons laut Vorgabe angeordnet werden. 
Dieser letzte Punkt wird über RestoreDesktop erledigt und es gibt eine separate Doku dazu auf dem VoBS. 

1.3. Überblick über unsere Zuordnung des Hintergrundbildes 

Über eine Gruppenrichtlinie wird für jede Benutzergruppe (grpSchueler, grpLehrer, grpDirektion, …) das im Verzeichnis 
C:\WINDOWS\Hintergrund\ liegende passende Hintergrundbild (Schulnetz_S.bmp, Schulnetz_L.bmp, …)  zugeordnet. 

1.3.1. Benutzergruppen 

Es gibt Hintergrundbilder für folgende Benutzergruppen bzw. Benutzer:  
 

 grpSchueler    - Schulnetz_S.bmp   
 grpLehrer  - Schulnetz_L.bmp 

 grpDirektion    - Schulnetz_D.bmp 

 grpECDL  - Schulnetz_E.bmp 

 grpKurs  - Schulnetz_Ku.bmp 

 grpSonstige  - Schulnetz.bmp 

 Kustos  - Schulnetz_K.bmp 

 Administrator  - Schulnetz_A.bmp 

1.3.2. Auflösungen 

Für folgende Auflösungen sind  Unterordner mit passenden Hintergrundbildern vorhanden. 

 1024 x 768  

 1280 x 1024  

 1680 x 1050 

 1920 x 1080 

 1366 x 768 - Notebook 
 1600 x 900 – Notebook 

1.3.3. Manuelle Anpassung pro PC 

 Kopieren der Hintergrundbilder:   
Die Hintergrundbilder  im Unterordner  
der gewünschten Auflösung müssen nach 
c:\WINDOWS\Hintergrund kopiert werden.  

 Einstellen der Auflösung: 
Es muss manuell die passende Auflösung  
in Windows (z.B: 1920x1080) eingestellt werden. 
 
Beim nächsten Anmelden sollten dann Auflösung und  
Hintergrundbild mit dem Monitor zusammenpassen.  

2. Automatisierung mit dem Skript „identifyMonitor.bat“ 

2.1. Überblick über Abläufe 

Beim Start des Domänen-PCs wird mittels „StartPC.bat“ das Skript „identifyMonitor.bat“ aufgerufen, sofern dies vom 
Administrator „eingeschaltet“ wurde.  „identifyMonitor.bat“ geht seine Liste mit den eingetragenen Monitorkennungen 
durch und ermittelt jeweils über Windows, ob der aktuell angeschlossene Monitor diese Kennung hat. Bei 
Übereinstimmung  werden die im Skript bei dieser Kennung hinterlegten Werte  dazu verwendet, das richtige 
Hintergrundbild zu kopieren und die Auflösung entsprechend einzustellen. 

2.2. Ermitteln der Monitorkennung 

2.2.1. Hardware-ID mittels Gerätemanager 

Jeder Monitor hat unter Windows  
eine Hardware-ID. Diese kann im  
Gerätemanager angeschaut werden. 
Re. Maus auf „PnP-Monitor“ -  
Eigenschaften – Details -  
Hardware-IDs 
 
Kennung ist  hier:  „ACR00D3“ 
(Acer B243HL  -24“-Zoll  
von Ausschreibung  2010) 
 

2.2.2. Hardware-ID mittels Skript und Hilfsprogramm „devcon“ 

Das Skript „getMonitorType.bat“ (als Administrator!) starten:  
Rechte Maus  - „Als Administrator ausführen“  
 
Als Basis für das Skript wird das Tool „Devcon.exe“ von 
Microsoft verwendet, das sozusagen ein textorientierter 
„Gerätemanager“ auf Commandlinebasis ist. 
 
Der Befehl  „DEVCON.EXE hwids =MONITOR“  (in einem Administrator-CMD.exe wegen UAC) zeigt ähnliches. 
Auf ähnliche Weise ermittelt auch „identifyMonitor.bat“ den angeschlossenen Monitor. 

2.3. Eintragen der Monitorkennung im Skript 

Die Monitorkennung samt zugehörigen Werten muss im Skript eingetragen sein: 
 

 

Die Monitore der OEBS-Ausschreibungen der Jahre 2002 bis 2011 sind bereits eingetragen.  
Bei im Skript fehlenden Monitoren einen ähnlichen (passenden) Block kopieren und Werte entsprechend ergänzen! 

2.4. Ausnahmen für bestimmte Computer 

Bei bestimmten Computern ist eine automatische Einstellung der Auflösung entsprechend dem angeschlossenen Monitor 
nicht gewünscht, etwa bei Lehrergeräten mit angeschlossenem Beamer.  Dafür gibt es die beiden folgenden Varianten. 

2.4.1. Sofortiges Beenden des Skripts 

Durch Eintragen des Computernamens bei der „Sofort-Raus“-Sektion wird das Skript sofort beendet, ohne die Auflösung 
zu ändern oder die Hintergrundbilder zu kopieren. Dies muss  dann bei Bedarf  manuell am PC erfolgen. 

::================== 

:: 24 Zoll TFT 

::================== 

:: 

::-----2010-ACER-24-Zoll--------------------------------------- 

set DESCR=2010-ACER-24-Zoll 

set MODELL=ACR00D3 

set MYWIDTH=1920 

set MYHEIGHT=1080 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

Beliebige Bezeichnung zur Wiedererkennung 
 

Die Hardware-ID aus dem Gerätemanager 
 

Gewünschte  Auflösung: hier 1920 x 1080 
(muss der Monitor natürlich können) 

 

Bildwiederholfrequenz: bei TFT 60 Hz 
Bei CRT-Röhren  min. 75 Hz wg. Flimmern 

2.4.2. Verwendung einer fixen Auflösung 

Durch Eintragen des Computernamens bei der „Zwangsauflösung“-Sektion wird unabhängig vom angeschlossenen 
Monitor die hier vorgegebene Auflösung verwendet. Das Skript stellt dann diese Auflösung  ein und kopiert die 
Hintergrundbilder.  Natürlich muss der Monitor diese Auflösung unterstützen. 
 

 

Das Sprungziel mit den Vorgabe-Werten ist weiter unten im Skript definiert: 
 

 

Weiters gibt es noch die beiden Sprungziele  „SETRES_1280x1024_max“ und „SETRES_1920x1080_max“, wobei der 
Parameter MYFREQUENCY=max die für den entsprechenden Monitor beste Wiederholfrequenz einstellen soll.  
 

Nach diesem Muster können bei Bedarf weitere Einstellungen  hinzugefügt werden. 
 

::======================================================== 

::--- Einstellungen nach COMPUTERNAME ------------------------------------- 

::======================================================== 

::------------------------------------------------------------------------------ 

::------Sofort-RAUS-bei BEAMER oder Multimonitorbetrieb-- 

::------------------------------------------------------------------------------ 

:: Fuer folgende PCs dieses Programm sofort verlassen ... 

::----------------------------------------------------------------- 

::  (etwa Lehrer-PCs mit Beamer trotz 19 Zoll Monitor oder mit 2 Monitoren) 

::  (wo die Aufloesung manuell am PC eingestellt werden soll              ) 

:: 

if /i %COMPUTERNAME%==PCDIR999        goto :ende 

if /i %COMPUTERNAME%==PC99_mit_Beamer goto :ende 

:: 

::-----Sofort-RAUS-ENDE-------------------------------------------------------- 

::------------------------------------------------------------------------------ 

::------Zwangsaufloesung bei BEAMER - Betrieb------------------ 

::------------------------------------------------------------------------------------------------------ 

:: Fuer folgende PCs unabhaengig vom Monitor angegebene Aufloesung einstellen 

::------------------------------------------------------------------------------------------------------- 

::  (etwa Lehrer-PCs mit Beamer trotz 19 Zoll Monitor ) 

:: 

if /i %COMPUTERNAME%==PC_BEAM_1024    goto :SETRES_1024x768_60 

if /i %COMPUTERNAME%==PC_HDTV                goto :SETRES_1920x1080_max 

:: 

::------Zwangsaufloesung-ENDE--------------------------------------------------- 

:: ========MANUELLE EINSTELLUNG================================================= 

::  

::------------------Manuelle Einstellung -1024 x 768 - 60Hz -------- 

:SETRES_1024x768_60 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=60 

set MONITOR=Admin-Vorgabe-%MYWIDTH%x%MYHEIGHT% 

goto MONITOR_found 

::------------------------------------------------------------------ 

3. Verwendung des Skripts „identifyMonitor.bat“ 

3.1. Wo Liegt das Skript und die zugehörigen Programme 

Dass Skript liegt in \\server\netlogon\utilities\setresolution   
 
 

 

Tools: 
 
devcon-i86.exe               -  Gerätemanager – cmdline 32bit 
                                              von Microsoft 
dccmd.exe                       -  Display Changer 
                                              Einstellen der Auflösung – cmdline 
                                              von 12noon, Stefan Tucker 
dc.exe                               -  Einstellen der Auflösung grafisch 
                                              von 12noon, Stefan Tucker 
 
WizApp.exe                   -  gibt der Steuerbatch zum  
                                           Ein-/Ausschalten von IdentifyM 
                                           die grafische Oberfläche 
DO-Setresolution.cmd   -  IdentifyMonitor via StartPC.bat  
                                           aktivieren /deaktivieren 
 
getMonitor.bat               -  Batch ermittelt Monitorkennung 
 
identifyMonitor.bat        -  Hauptskript                                
 

 

3.2. Starteintrag in „StartPC.bat“ im NETLOGON 

Beim Systemstart eines Domänen-PCs wird in der Musterinstallation der Regional- 
betreuer das Startskript „StartPC.bat“ aufgerufen.  
Darin ist u.a. folgender Eintrag: (if exist … - einzeilig!) 
 

 
 
 
 
 
 
 
 
 
 
 

Wenn also im Netlogon-Verzeichnis die Datei “DO-SETRESOLUTION.txt” vorhanden ist, wird „IdentifyMonitor.bat“ 
aufgerufen und somit die Auflösungseinstellung auf den Clients bei deren Systemstart mit Systemrechten durchgeführt. 

3.3. SetResolution ein- bzw. ausschalten 

Die Batch „Do-SetResolution.cmd“  in \\server\netlogon\utilities\Setresolution   
erzeugt bzw. löscht über eine kleine grafische Benutzeroberfläche die  
Triggerdatei „DO-SETRESOLUTION.txt“ im Netlogon-Verzeichnis, wenn  
„Bei jedem Start  Grafikauflösung einstellen“ an- bzw. abgewählt wird.   
 
Sie kann auch über den Link „Administration“ auf dem Serverdesktop  
mit „Auflösung für Schüler-PCs umstellen“ direkt  
aufgerufen werden. Dabei erscheint/verschwindet  
auf dem Server-Desktop als Erinnerung eine Infodatei. 
 
 
 
 
 

REM   In STARTPC.BAT  

REM   --------------------- 

 

rem -----Set_ScreenResolution-----------------------------------------

- 

Aufruf einer Batchdatei zum Setzen der Bildschirm-Aufloesung  

rem abhaengig vom Monitormodell 

rem wenn die Datei DO-SETRESOLUTION.txt da ist 

rem ------------------------------------------------------------------

if exist \\server\netlogon\DO-SETRESOLUTION.txt call    

          
\\server\netlogon\Utilities\SetResolution\identifyMonitor.bat 

rem ---------------------------------------- -------------------------

------------------------ 

file://server/netlogon/utilities/setresolution
file://server/netlogon/utilities/Setresolution

Das „Einstellen der Grafikauflösung“ auf den Clients mit dem Kopieren der Hintergrundbilder ist natürlich nicht bei jedem 
Start nötig. Es kostet mit den eingebauten Verzögerungen etwa 9 Sekunden und kann abgeschaltet werden, wenn jeder 
PC nach dem Clonen und der Domänenanbindung die gewünschten Einstellungen einmal erhalten hat. 

3.4. Log-Datei bei „unbekanntem“ Monitor 

Hängt während der Ausführung von „identifyMonitor.bat“ ein Monitor am PC, dessen Kennung nicht im Skript 
eingetragen ist, so wird auf diesem PC kurz eine Information mit den Daten des Monitors angezeigt und dann auf die 
Freigabe \\server\logs\monitor\ unter dem Dateinamen „MonitorInfo-COMPUTERNAME.txt“ kopiert, wobei der 
tatsächliche Computername verwendet wird. Dadurch hat der Administrator bequem alle Information über fehlende 
Einträge in Sichtweite.  Die Freigabe \\server\logs  muss existieren und das System Schreibrechte darin haben. 
 

Wichtig ist, dass auf den PCs die korrekten Grafiktreiber installiert sind, da ansonsten nicht wirklich hilfreiche 
Informationen erscheinen, wie das linke Beispiel zeigt. 
 

Auszug aus „MonitorInfo-PC22.txt  Auszug aus „MonitorInfo-PC01.txt 
-----------------------------------------------

-- 

Info von IdentifyMonitor  

-----------------------------------------------

-- 

Problem: --> Monitorname nicht in Batch 

gefunden 

 

---Name des Computers------------------------ 

PC22 

 

---Name(n) des Monitors in der Registry------ 

        Monitor\Default_Monitor 

 

---maximale Werte des Monitors--------------- 

Display: \\.\DISPLAY1 

Width:  1920 

Height:  1440 

Color:  32 

Frequency: 1 

 

(Moeglicherweise sind die Werte nicht korrekt.  

Bitte manuell einstellen und ueberpruefen! 

 ------------------------------------------------

- 

Info von IdentifyMonitor  

------------------------------------------------

- 

Problem: --> Monitorname nicht in Batch gefunden 

 

---Name des Computers------------------------ 

PC01 

 

---Name(n) des Monitors in der Registry------ 

        Monitor\SAM009A 

 

---maximale Werte des Monitors--------------- 

Display: \\.\DISPLAY1 

Width:  1024 

Height:  768 

Color:  32 

Frequency: 1  

 

(Moeglicherweise sind die Werte nicht korrekt.  

Bitte manuell einstellen und ueberpruefen! 

 

Bei PC22 sind die Werte für die Auflösung  falsch.                                Der falsche Wert für Frequenz wird durch einen 
Tatsächlich ist es ein 19“ Monitor mit 1280 x 1024.                              Monitorumschalter verursacht. 
Ursache ist der fehlende Grafikkartentreiber, der nur                          Und genau wegen solcher Fehlerkennungen wäre 
als „Standard-Grafik-Treiber“ funktioniert und den                              eine vollautomatische Erkennung problematisch. 
Monitor nicht richtig erkannte.  Nach der Installation des                   Da soll der Admin ruhig auch was tun ;-)) 
ATI-Grafik-Treibers wurde der Monitor korrekt erkannt! 

file://server/logs/monitor/
file://server/logs

3.5. Manuelles Ausführen des Skripts 

Die automatische Ausführung innerhalb des Systemstarts des Domänen-PCs ist völlig problemlos, da das Programm dort 
unter der Berechtigung „System“ mit vollen Rechten läuft. 
 

Das manuelle Ausführen von „identifyMonitor.bat“ als Benutzer kann jedoch eine Menge Probleme bereiten, wenn unter 
Windows 7 die Benutzerkontensteuerung (User-Access-Control  = UAC) eingeschaltet ist.  Das Hilfsprogramm 
„devcon.exe“ von Microsoft, das im Skript verwendet wird,  benötigt für den Zugriff auf den „Gerätemanager“ 
Administratorrechte.  
 
 
 
 
Da für jeden Monitor dieser Teil erneut aufgerufen wird,  
nervt das Programm dann mit unzähligen UAC-Abfragen  
und bringt dennoch die Antwort nicht auf den Bildschirm.  
 
Ebenso erscheint beim Kopieren des Hintergrundes  
nach c:\windows\Hintergrund  die Meldung „Zugriff verweigert“  
 
Daher muss das Programm „identifyMonitor.bat“ gleich zu Beginn mit „Als Administrator ausführen“ gestartet werden! 
 
Am sichersten gelingt dies, wenn man als 
 Administrator angemeldet ist und eine  
Admin-CMD geöffnet wird, in der dann  
die Batch „identifyMonitor.bat“ ausgeführt wird. 
 
 
 
 
 
Und so soll es dann aussehen: 
 

 
 
 
 

So funktioniert es NICHT ! 

D    re. Maus – und  „Als Administrator ausführen“ 

4. Anhang mit Listings 
Es folgt eine Auflistung der Skripte (zu Kopierzwecken) 

4.1. getMonitorType.bat        
  

@echo off 

goto BEGINN 

::============================================================================== 

::------------------------------------------------------------------------------ 

:: getMonitorType - Monitormodell ermitteln - v. 1.2   - Juni 2011  

:: ----------------------------------------------------------------------------- 

:: 

:: Info: IT-Regionalbetreuer Vlbg. - Helmut Pajnik 

::------------------------------------------------------------------------------ 

:: Diese Batch ermittelt das Monitormodell mit devcon.exe 

:: Das Ergebnis wird angezeigt 

:: ----------------------------------------------------------------------------- 

:: --Alternative unter Windows -- 

::  

:: Gerätemanger oeffnen und z.B. das Monitormodell auswählen  

:: - Eigenschaften der Monitore  

:: - unter Details die Geräteinstanzkennung auswaehlen. 

::   (ergibt z.B: DISPLAY\MAX077B\4&3B42C36E&0&80861100&00&02 

::         d.h:    DISPLAY 

::                 MAX077B  = Modellnummer           

::============================================================================== 

 

::============================================================================== 

:BEGINN 

cls 

echo --INFO--------------------------------------------------------------- 

echo Diese Batch ermittelt das angeschlossene Monitormodell des PCs. 

echo.  (verwendet devcon.exe und benoetigt Administratorrechte) 

echo - starten mit -- ALS ADMINISTRATOR AUSFUEHREN - bei Win7 wegen UAC  

echo.- oder besser -- in einer Administrator - cmd.exe 

echo.  !!! Andernfalls erscheint keine Info zum Monitor !!! 

echo.          

echo Remoteaufruf mit Parameter: z.B: getMonitorType PC15 

echo.                                 zeigt Monitormodell von PC15 

echo --INFO-ENDE----------------------------------------------------------  

::============================================================================== 

 

::------------------------------------------------------------------ 

::--- Initialisierung----------------------------------------------- 

::------------------------------------------------------------------ 

set STARTPATH=%~dp0 

if %STARTPATH:~-1%==\ set STARTPATH=%STARTPATH:~0,-1% 

rem Pfad sichern und ev.vorhandenen Backslash wegschneiden 

 

if /i %PROCESSOR_ARCHITECTURE%==x86   set DEVCONPROG=devcon-i86.exe 

if /i %PROCESSOR_ARCHITECTURE%==amd64 set DEVCONPROG=devcon-amd64.exe  

if /i %PROCESSOR_ARCHITECTURE%==IA64  goto :EOF 

:: 

rem -------- 

rem x86   -- 32bit-Betriebssystem (Intel und AMD) 

rem amd64 -- 64bit-Betriebssystem (Intel und AMD) 

rem ia64  -- 64bit-Betriebssystem (Itanium Prozessor) 

::============================================================================== 

 

set MYPC=SELF 

if not "%1"=="" set MYPC=%1 

echo Ermittle angeschlossenen Monitor 

 

if %MYPC%==SELF ( 

echo auf diesem PC: %COMPUTERNAME% 

  "%STARTPATH%\%DEVCONPROG%" hwids =MONITOR|find /i "monitor"|find /i /v "Name" 

) else ( 

echo auf Remote-PC: %MYPC%:  

echo Natuerlich werden auf dem angegebenen PC entsprechende RECHTE benoetigt. 

echo. 

  "%STARTPATH%\%DEVCONPROG%" -m:\\%1 hwids =MONITOR |find /i "monitor"|find /i /v "Name" 

) 

echo. 

echo (Falls hier kein Monitor angezeigt wird fehlen Administratorrechte!) 

echo. 

pause 

4.2.  identifyMonitor.bat 
 

@echo off 

goto BEGINN 

::=============================================================================== 

::------------------------------------------------------------------------------- 

:: Monitormodell ermitteln und optim. Aufloesung einstellen - v. 1.3  - Juni 2011  

:: ------------------------------------------------------------------------------ 

:: 

:: Info: IT-Regionalbetreuer Vlbg. - Helmut Pajnik 

::------------------------------------------------------------------------------- 

:: Diese Batch ermittelt das Monitormodell mittels devcon. 

:: Es wird dann die angegebene Wunsch-Aufloesung eingestellt. 

:: Ebenso werden die passenden Hintergrundbilder aus  

:: c:\windows\hintergund\1024x768 (bzw. jeweilige Aufloesung) 

:: nach c:\windows\hintergrund kopiert (und per Gruppenrichtlinie zugewiesen). 

:: Derzeit unterstuetzte Hintergrundaufloesungen: 

::    - 1024 x 768  

::    - 1280 x 1024  

::    - 1680 x 1050 

::    - 1920 x 1080 

::    - 1366 x 768 - Notebook 

::    - 1600 x 900 - Notebook 

:: ------------------------------------------------------------------------------ 

:: --Vorarbeit unter Windows fuer zusaetzliche Monitor-Modelle-- 

:: Im Geraetemanager die Modellnummer des angeschlosssenen Monitors suchen. 

::  

:: Dazu den Gerätemanger oeffnen und das Monitormodell auswählen  

:: - Eigenschaften der Monitore  

:: - unter Details die Geräteinstanzkennung bzw. Hardware-ID auswaehlen. 

::   ergibt z.B: DISPLAY\MAX077B\4&3B42C36E&0&80861100&00&02 

::         d.h:    DISPLAY 

::                 MAX077B  = Modellnummer 

::                  

:: 

:: --In dieser Batch-- 

:: In dieser Batch muss dann ein zusaetzlicher Block passend erstellt werden 

:: Am besten einen vorhandennen kopieren und den Namen anpassen! 

:: Es erfolgt dann die gezielte Abfrage auf das Modell! 

:: zb:  

::    -----Maxdata-19-Zoll-alle-TFT-------------------------------- 

::    set DESCR=Belinea-Maxdata alle 19 Zoll 

::    set MODELL=MAX07 

::    set MYWIDTH=1280 

::    set MYHEIGHT=1024 

::    set MYFREQUENCY=60 

::    ---weiter-wie-immer--- 

:: 

:: Bemerkung: Hier wurde nur MAX07 statt MAX077B genommen 

::            wegen einiger nahezu baugleicher Typen  

::             

:: 

:: --Infodatei----- 

:: Bei unbekanntem Monitor wird beim STart des PCs eine Infodatei  

:: mit den  ermittelten Werten des Monitors ins Verzeichnis 

:: \\server\logs\monitor kopiert. 

:: Diese Daten sollen bei der Eintragung helfen, 

:: jedoch ohne Gewaehr.  

:: Ein Splitter kann bspw. falsche Werte liefern! 

::=============================================================================== 

 

 

::============================================================== 

::----------SUBROUTINE-CHECKSTRING----------------------------- 

::------------------------------------------------------------- 

:: aendert Varaible %MONITOR% 

::------------------------------------------------------------- 

:CheckString 

"%STARTPATH%\%DEVCONPROG%" hwids * |find /I /C %1% >NUL 

rem "%STARTPATH%\%DEVCONPROG%" hwids * |find /I  %1%  

if NOT ERRORLEVEL 1  set MONITOR=%MODELL% 

rem nur wenn Monitor 1 x vorkommt 

rem ping /n 2 127.0.0.1 > nul 

goto :EOF 

::-------------------------------------------------------------- 

::----------END-SUBROUTINE-CHECKSTRING-------------------------- 

::=============================================================== 

 

 

:BEGINN 

:: 

::============================================================================== 

::------------------------------------------------------------------ 

::--- Initialisierung----------------------------------------------- 

::------------------------------------------------------------------ 

:: 

set SERVERLOGDIR=\\server\logs 

rem set SERVERLOGDIR=c: 

rem fuer MonitorInfo-Datei bei unbekanntem Monitor 

set MONITOR=UNKNOWN 

set STARTPATH=%~dp0 

if %STARTPATH:~-1%==\ set STARTPATH=%STARTPATH:~0,-1% 

rem Pfad sichern und ev.vorhandenen Backslash wegschneiden 

:: 

if /i %PROCESSOR_ARCHITECTURE%==x86   goto :x86 

if /i %PROCESSOR_ARCHITECTURE%==amd64 goto :amd64 

if /i %PROCESSOR_ARCHITECTURE%==IA64  goto :EOF 

::  

::----richtige-Version-zuordnen-------------------- 

:: 

::-------x86-----32bit-Betriebssystem (Intel und AMD) 

:x86 

set DEVCONPROG=devcon-i86.exe 

set SETRESPROG=dccmd.exe 

goto :END_INIT 

:: 

::-------amd64---64bit-Betriebssystem (Intel und AMD) 

:amd64 

set DEVCONPROG=devcon-amd64.exe  

set SETRESPROG=dc64cmd.exe 

goto :END_INIT 

:: 

::------ia64  -- 64bit-Betriebssystem (Itanium Prozessor) 

:ia64 

rem --- haben wir nicht --- 

goto :EOF 

:: 

:END_INIT 

::------------------------------------------------------------------ 

::--- Initialisierung-ENDE------------------------------------------ 

::============================================================================== 

 

 

 

::============================================================================== 

::--- Einstellungen nach COMPUTERNAME ------------------------------------------ 

::============================================================================== 

:: 

::------------------------------------------------------------------------------ 

::------Sofort-RAUS-bei BEAMER oder Multimonitorbetrieb------------------------- 

::------------------------------------------------------------------------------ 

:: Fuer folgende PCs dieses Programm sofort verlassen ... 

::----------------------------------------------------------------- 

::  (etwa Lehrer-PCs mit Beamer trotz 19 Zoll Monitor oder mit 2 Monitoren) 

::  (wo die Aufloesung manuell am PC eingestellt werden soll              ) 

:: 

if /i %COMPUTERNAME%==PCDIR999        goto :ende 

if /i %COMPUTERNAME%==PC99_mit_Beamer goto :ende 

:: 

::-----Sofort-RAUS-ENDE--------------------------------------------------------- 

:: 

:: 

::------------------------------------------------------------------------------ 

::------Zwangsaufloesung bei BEAMER - Betrieb----------------------------------- 

::------------------------------------------------------------------------------ 

:: Fuer folgende PCs unabhaengig vom Monitor angegebene Aufloesung einstellen 

::---------------------------------------------------------------------------- 

::  (etwa Lehrer-PCs mit Beamer trotz 19 Zoll Monitor ) 

:: 

if /i %COMPUTERNAME%==PC_BEAM_1024    goto :SETRES_1024x768_60 

if /i %COMPUTERNAME%==PC_HDTV         goto :SETRES_1920x1080_max 

if /i %COMPUTERNAME%==PCHPTV          goto :SETRES_1920x1080_max 

rem if /i %COMPUTERNAME%==PCHPTV          goto :SETRES_1280x1024_max 

:: 

::------Zwangsaufloesung-ENDE--------------------------------------------------- 

::============================================================================== 

::============================================================================== 

::--- Einstellungen nach MONITORMODELL------------------------------------------ 

::============================================================================== 

:: 

echo ---------------------------------------------------------- 

echo Ermittle Monitormodell der Ausschreibungen 2002 bis 2011 

echo ---------------------------------------------------------- 

:: 

::================== 

:: CRT- 17 ZOLL 

::================== 

:: 

::-----2001-Siemens-CRT-17-Zoll-------------------------------- 

set DESCR=2001-Siemens-CRT-17-Zoll 

set MODELL=SIE2001 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=75 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2001-EIZO-CRT-17-Zoll----------------------------------- 

set DESCR=2001-EIZO-CRT-17-Zoll 

set MODELL=EIZ1008 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=75 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2002-Samsung-CRT-17-Zoll-------------------------------- 

set DESCR=2002-Samsung-CRT-17-Zoll 

set MODELL=SAM0027 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=75 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::================== 

:: 15 Zoll TFT 

::================== 

:: 

::-----Maxdata-TFT-15-Zoll-generic----------------------------- 

set DESCR=Maxdata-Belinea TFT- alle 15-Zoll  

set MODELL=MAX05 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%"   ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2003-Maxdata-TFT-15-Zoll-------------------------------- 

set DESCR=2003-Maxdata-TFT-15-Zoll 

set MODELL=MAX0613 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----Samsung-TFT-15-Zoll-------------------------------- 

set DESCR=Samsung-Syncmaster-TFT-15-Zoll 

set MODELL=SAM009A 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

:: 

::================== 

:: 17 Zoll TFT 

::================== 

:: 

::-----Maxdata-17-Zoll-alle-TFT-------------------------------- 

set DESCR=Belinea-17 Zoll 

set MODELL=MAX06C2 

set MYWIDTH=1280 

set MYHEIGHT=1024 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----Belinea-17-Zoll-Röhrenmonitor--------------------------- 

set DESCR=Belinea-17 Zoll 

set MODELL=MAX0B 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=75 

echo Teste Kennung: "%MODELL%"   ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

:: 

::================== 

:: 19 Zoll TFT 

::================== 

:: 

::-----Maxdata-19-Zoll-alle-TFT-------------------------------- 

set DESCR=Belinea-Maxdata alle 19 Zoll 

set MODELL=MAX07 

set MYWIDTH=1280 

set MYHEIGHT=1024 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%"   ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2009-ACER-19-Zoll--------------------------------------- 

set DESCR=2009-ACER-19-Zoll-LCD-B193  

set MODELL=ACR001D 

set MYWIDTH=1280 

set MYHEIGHT=1024 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2009b-ACER-19-Zoll-------------------------------------- 

set DESCR=2009b-ACER-19-Zoll-LCD- 

set MODELL=ACR004B 

set MYWIDTH=1280 

set MYHEIGHT=1024 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

:: 

::================== 

:: 22 Zoll TFT 

::================== 

:: 

::-----2009-ACER-22-Zoll--------------------------------------- 

set DESCR=2009-ACER-22-Zoll 

set MODELL=ACR0018 

set MYWIDTH=1680 

set MYHEIGHT=1050 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2010-ACER-22-Zoll--------------------------------------- 

set DESCR=2010-ACER-22-Zoll 

set MODELL=ACR0020 

set MYWIDTH=1680 

set MYHEIGHT=1050 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2011-ACER-22-Zoll--------------------------------------- 

set DESCR=2011-ACER-22-Zoll 

set MODELL=ACR0169 

set MYWIDTH=1680 

set MYHEIGHT=1050 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

:: 

::================== 

:: 24 Zoll TFT 

::================== 

:: 

::-----2010-ACER-24-Zoll--------------------------------------- 

set DESCR=2010-ACER-24-Zoll 

set MODELL=ACR00D3 

set MYWIDTH=1920 

set MYHEIGHT=1080 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

:: 

::================== 

:: NOTEBOOKS 

::================== 

:: 

::-----2010-Lenovo-T-510-Basic-------------------------------------- 

set DESCR=2010-Notebook-Lenovo-T-510-Basic 1366x768 

set MODELL=LEN40B0 

set MYWIDTH=1366 

set MYHEIGHT=768 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2010-Lenovo-T-510-HI-RES-------------------------------------- 

set DESCR=2010-Notebook-Lenovo-T-510-Hi_RES 1600x900 

set MODELL=LEN40B1 

set MYWIDTH=1600 

set MYHEIGHT=900 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2011-HP-ProBook-6550b-BASIC---------------------------- 

set DESCR=2011-Notebook-HP-ProBook-6550b-Basic 1366x768 

set MODELL=SEC4351 

set MYWIDTH=1366 

set MYHEIGHT=768 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

::-----2011-HP-ProBook-6550b-HI-RES---------------------------- 

set DESCR=2011-Notebook-HP-ProBook-6550b-HI-RES 1600x900 

set MODELL=AUO01EE 

set MYWIDTH=1600 

set MYHEIGHT=900 

set MYFREQUENCY=60 

echo Teste Kennung: "%MODELL%" ---%DESCR%  

call :CheckString "MONITOR\%MODELL%" 

if not %MONITOR%==UNKNOWN goto MONITOR_found 

::------------------------------------------------------------- 

:: 

 

 

 

 

::================================================================ 

::-------MONITOR NOT-FOUND------------------------ 

::================================================ 

rem Monitor-MODELL in obiger Liste nicht angegeben  

:: 

echo. 

echo --------------------------------------- 

echo Monitor nicht angegeben oder gefunden 

echo --------------------------------------- 

echo. 

rem ============================================== 

rem -- Ausgabefenster mit INFO ------------------- 

rem ============================================== 

:: 

echo.> %temp%\Monitorinfo.txt 

( 

echo ------------------------------------------------- 

echo Info von IdentifyMonitor  

echo ------------------------------------------------- 

echo. 

echo Problem: --^> Monitorname nicht in Batch gefunden 

echo. 

echo Diese Batch stellt die optimale Aufloesung ein  

echo und kopiert den passenden Bildschirmhintergrund. 

echo Wegen fehlender Daten in der Batchdatei ist dies 

echo fuer diesen Monitor derzeit nicht moeglich.  

echo. 

echo ---Name des Computers------------------------ 

echo.%COMPUTERNAME% 

echo. 

echo ---Name^(n^) des Monitors in der Registry------ 

"%STARTPATH%\%DEVCONPROG%" hwids =MONITOR|find /i "monitor"|find /i /v "Name" 

echo. 

echo ---maximale Werte des Monitors--------------- 

"%STARTPATH%\%SETRESPROG%" -max -test |find /V "Test" 

echo ^(Moeglicherweise sind die Werte nicht korrekt. 

echo Bitte manuell einstellen und ueberpruefen!^) 

echo --------------------------------------------- 

echo. 

echo Die Batchdatei ^(im NETLOGON^) 

echo."%0" 

echo anhand der obenstehenden Daten laut  

echo folgendem Muster sinnvoll ergaenzen: 

echo. 

echo ------------------------------------Beispiel---- 

echo.    -----Maxdata-19-Zoll-TFT-------------------- 

echo.    set DESCR=Belinea-Maxdata alle 19 Zoll 

echo.    set MODELL=MAX07FF 

echo.    set MYWIDTH=1280 

echo.    set MYHEIGHT=1024 

echo.    set MYFREQUENCY=60 

echo.    - - - weiter wie immer  - - - 

echo ------------------------------------------------ 

echo. 

echo -- ODER -- 

echo wenn fuer diesen PC etwa wegen Beamer- oder 

echo Dual-Monitorbetrieb keine automatische  

echo Aufloesungseinstellung gewuenscht ist, dann  

echo am Beginn der Batch bei -sofort raus- folgende  

echo Zeile einfügen: 

echo. 

echo if /i %%COMPUTERNAME%%==%COMPUTERNAME%   goto :ende 

echo. 

echo. 

echo --------------------------------------------- 

echo --- Zusatzinfo: Ausgaenge der Grafikkarte --- 

echo --------------------------------------------- 

echo. 

"%STARTPATH%\%SETRESPROG%" -listmonitors 

echo. 

echo. 

echo ---------------------------------------------------------------- 

echo ---weitere Infos auf VoBS oder bei den IT-Regionalbetreuern----- 

echo --------------------------------------------- H. Pajnik -------- 

echo. 

) >> %temp%\Monitorinfo.txt 

start "INFO" "notepad.exe" %temp%\Monitorinfo.txt 

if not exist "%SERVERLOGDIR%\." ( 

echo Logverzeichnis %SERVERLOGDIR% nicht gefunden 

) else ( 

echo Kopiere Info ins Logverzeichnis %SERVERLOGDIR% ... 

if not exist "%SERVERLOGDIR%\monitor\." md "%SERVERLOGDIR%\monitor" 

copy "%temp%\Monitorinfo.txt" "%SERVERLOGDIR%\Monitor\Monitorinfo-%COMPUTERNAME%.txt" /Y 

) 

REM  

REM ---Kurze-Verschnaufpause-------------------- 

ping -n 8 127.0.0.1 >nul 

rem ============================================== 

rem -- Ausgabefenster mit INFO --ENDE------------- 

rem ============================================== 

::-------NOT-FOUND-END---------------------------- 

::================================================================ 

:: 

goto :ende 

 

 

:: ========MANUELLE EINSTELLUNG================================================= 

::  

::------------------Manuelle Einstellung -1024 x 768 - 60Hz -------- 

:SETRES_1024x768_60 

set MYWIDTH=1024 

set MYHEIGHT=768 

set MYFREQUENCY=60 

set MONITOR=Admin-Vorgabe-%MYWIDTH%x%MYHEIGHT% 

goto MONITOR_found 

::------------------------------------------------------------------ 

::  

::------------------Manuelle Einstellung -1280 x 1024 max Hz-------- 

:SETRES_1280x1024_max 

set MYWIDTH=1280 

set MYHEIGHT=1024 

set MYFREQUENCY=max 

set MONITOR=Admin-Vorgabe-%MYWIDTH%x%MYHEIGHT% 

goto MONITOR_found 

::------------------------------------------------------------------ 

::  

::------------------Manuelle Einstellung -1920 x 1080 max Hz-------- 

:SETRES_1920x1080_max 

set MYWIDTH=1920 

set MYHEIGHT=1080 

set MYFREQUENCY=max 

set MONITOR=Admin-Vorgabe-%MYWIDTH%x%MYHEIGHT% 

goto MONITOR_found 

::----------------------------------------------------------- 

:: 

:: ========MANUELLE EINSTELLUNG-ENDE============================================ 

 

 

::---------FOUND----------------------------------------------------------------------- 

:MONITOR_found 

echo. 

echo ---------------------------------------------- 

echo Das Monitormodell ist: %MONITOR% 

echo ---------------------------------------------- 

echo. 

echo Aufloesung von %MYWIDTH% x %MYHEIGHT% mit %MYFREQUENCY% Herz Wiederholfrequenz 

echo. 

REM ---Kurze-Verschnaufpause-------------------- 

ping -n 5 127.0.0.1 > nul 

:: 

:: 

REM ---Aufloesung-einstellen--------------------------------------------------------------- 

"%STARTPATH%\%SETRESPROG%" /width=%mywidth% /height=%myheight% /refresh=%myfrequency% /depth=max  

rem "%STARTPATH%\DCCmd.exe" /width=%mywidth% /height=%myheight% /refresh=max /depth=max  

rem refresh=max ermittelt Refresh selbst  - Funktion jedoch von uns nicht ausreichend getestet 

rem daher derzeit manuelle Angabe verwenden 

:: 

:: 

REM ---Hintergrundbild-loeschen-und-kopieren----------------------------------------------- 

del c:\windows\hintergrund\*.bmp /f/Q 

del c:\windows\hintergrund\*.jpg /f/Q 



 Hintergrundbild und Auflösung automatisch| Version 1.0 17 / 18 

 

xcopy "c:\windows\hintergrund\%MYWIDTH%x%MYHEIGHT%\*.*"  "c:\windows\hintergrund\*.*" 

/k/r/e/i/s/c/h/y 

:: 

::-------FOUND-END---------------------------------------------------------------------------- 

:: 

:ende  

:EOF 
 
 

4.3. DO_SETRESOLUTION.cmd 

Es folgt eine Variante, wie man mithilfe des Tools WizApp.exe mit einer Batch eine Benutzeroberfläche machen kann. 
 

@echo off 

SETLOCAL 

SETLOCAL ENABLEEXTENSIONS 

SETLOCAL ENABLEDELAYEDEXPANSION 

 

pushd %~dp0 

::                Verzeichnis, in dem die Batch aufgerufen wurde 

set STARTDIR1=%CD% 

set triggerfile=%STARTDIR1%\..\..\DO-SETRESOLUTION.txt 

cls 

 

:page1 

set waoutnum= 

if exist "%triggerfile%"       set waoutnum=0  

if NOT exist "%triggerfile%"   set waoutnum=1 

 

set watitle=Lokale Grafikaufloesung einstellen 

set wainput=Bei jedem Start Grafikaufloesung einstellen ;Grafikeinstellung wird NICHT aufgerufen 

 

set  watext=~Umstellen der Grafikaufloesung  

set watext2=~~ Beim Computerstart wird eine passende Grafikaufloesung   

set watext3= eingestellt und das Hintergrundbild geaendert! 

set watext4=~~~Bitte nur KURZZEITIG einschalten 

set watext5= 

 

set wabmp=%STARTDIR1%\Schulnetz.bmp 

if "%OS%"=="Windows_NT" set walabels=^&OK;^&Abbrechen;;^&Hilfe 

set wabat=%TEMP%\wabat.bat  

rem start "" /w "%startdir1%\wizapp" NOBACK  CL 

start "" /w "%startdir1%\wizapp"  RB 

if errorlevel 2 goto :page2 

rem  BTN_HILFE  

if errorlevel 1 goto :BTN_OK  

if errorlevel 0 goto :BTN_CANCEL  

 

:page2 

rem Hilfe 

set page=:page2 

set watext=Kurzinformation zu "SetResolution" 

set watext2=~ 

set watext3=~Beim Start des Schueler-PCs wird der  

set watext4=~angeschlossene Monitor ermittelt und die 

set watext5=~Aufloesung passend eingestellt. 

set watext6=~Ebenso wird das Hintergrundbild von  

set watext7=~C:\WINDOWS\HINTERGRUND\... passend kopiert 

set watext8=~~- 15 Zoll: 1024x768,   - 19 Zoll: 1280x1024 

set watext9=~- 22 Zoll: 1680x1050,  - 24 Zoll: 1920x1080 

set watext10=~- Notebook: 1366x768,  1600x900  

set watext11=~~Hinweis:  

set watext12=~    Fehlende Monitortypen koennen (sollten) im  

set watext13=~    Geraetemanager unter "Monitor - Details - 

set watext14=~    "Hardwarekennung" ermittelt und in der  

set watext15=~    Batchdatei "identifyMonitor" eingetragen werden! 

set watext16=~~- Infodatei bei unbekanntem Monitor unter  

set watext17=~   \\server\logs\monitor  

rem wafile contains the contents for the license box. 

rem set wafile=license.txt 

set wainput= 

rem The ^ is necessary to escape the ampersand on NT, 

rem The ampersand puts in a hotkey. 

rem if "%OS%"=="Windows_NT" set walabels=;I ^&Accept;;I ^&Disagree; 

rem if not "%OS%"=="Windows_NT" set walabels=;I &Accept;;I &Disagree; 

if "%OS%"=="Windows_NT" set walabels=^&Zur•ck;^&Beenden;;^&Datei sehen 

if not "%OS%"=="Windows_NT" set walabels=Zur•ck;Beenden;;^&Datei sehen 

rem start "" /w "%startdir1%\wizapp" FT 

start "" /w "%startdir1%\wizapp"  TB 

set MYCHOICE=%errorlevel% 

rem Don't forget to set wafile to empty -  

Seite 1:  Folgende Zeilen ergeben dieses Fenster 

Seite 2:  Folgende Zeilen ergeben dieses Fenster 

rem - if both wafile and wainput are set, wafile wins. 

set wafile= && rem And set the labels back to normal 

set walabels= 

set watext1= 

set watext2= 

set watext3= 

set watext4= 

set watext5= 

set watext6= 

set watext7= 

set watext8= 

set watext9= 
 

4.4. Beispiel einer Infodatei  “Monitorinfo-Computername.txt”  
 

 \\server\logs\monitor\ MonitorInfo-PC01.txt 
 

------------------------------------------------- 

Info von IdentifyMonitor  

------------------------------------------------- 

Problem: --> Monitorname nicht in Batch gefunden 

 

Diese Batch stellt die optimale Aufloesung ein  

und kopiert den passenden Bildschirmhintergrund. 

Wegen fehlender Daten in der Batchdatei ist dies 

fuer diesen Monitor derzeit nicht moeglich.  

 

---Name des Computers------------------------ 

PC01 

 

---Name(n) des Monitors in der Registry------ 

        Monitor\SAM009A 

 

---maximale Werte des Monitors--------------- 

Display:\\.\DISPLAY1 

Width:   1024 

Height:   768 

Color:     32 

Frequency: 60 

 

(Moeglicherweise sind die Werte nicht korrekt. 

Bitte manuell einstellen und ueberpruefen!) 

--------------------------------------------- 

Die Batchdatei (im NETLOGON) 

"\\server\netlogon\Utilities\SetResolution\identifyMonitor.bat" 

anhand der obenstehenden Daten laut  

folgendem Muster sinnvoll ergaenzen: 

 

------------------------------------Beispiel---- 

    -----Maxdata-19-Zoll-TFT-------------------- 

    set DESCR=Belinea-Maxdata alle 19 Zoll 

    set MODELL=MAX07FF 

    set MYWIDTH=1280 

    set MYHEIGHT=1024 

    set MYFREQUENCY=60 

    - - - weiter wie immer  - - - 

------------------------------------------------ 

-- ODER -- 

wenn fuer diesen PC etwa wegen Beamer- oder 

Dual-Monitorbetrieb keine automatische  

Aufloesungseinstellung gewuenscht ist, dann  

am Beginn der Batch bei -sofort raus- folgende  

Zeile einfügen: 

 

if /i %COMPUTERNAME%==PC01   goto :ende 

 

--------------------------------------------- 

--- Zusatzinfo: Ausgaenge der Grafikkarte --- 

--------------------------------------------- 

Monitor: PnP-Monitor (Standard) 

Device: \\.\DISPLAY1 

Adapter: ATI Radeon HD 4200 

(1024 x 768 x 32 bpp) frequency set in hardware up, attached, primary (0,0) 

---------------------------------------------------------------- 

---weitere Infos auf VoBS oder bei den IT-Regionalbetreuern----- 

--------------------------------------------- H. Pajnik -------- 

 


